連載

SUPER COLLIDER チュートリアル (4)
SUPER COLLIDER TUTORIALS (4)

美山 千香士
Chikashi Miyama
ケルン音楽舞踏大学
Hochschule für Musik und Tanz Köln

概要

本連載では、リアルタイム音響合成環境の SuperCollider(SC) の使い方を、同ソフトを作品創作や研究のために利用しようと考えている音楽家、メディア・アーティストを対象にチュートリアル形式で紹介する。
SuperCollider(SC) is a realtime programming environment for audio synthesis. This article introduces SC to musicians and media artists who are planning to utilize the software for their artistic creations and researches.

1. 今回の目標：SC でエフェクタを作成する



図 1. Mac のサウンドの設定

前回までは、SC を用いて音を生成する方法を学習してきたが、今回はマイクからの入力音をリアルタイムに加工する方法に焦点をあて、リング・モジュレーション、ディストーション、ピッチ・シフター、ディレイ、リバーブなどのエフェクトの作成をオムニバス的に紹介していく。ハウリングを避けるために、本稿のプログラムを実行する際には、ヘッドフォンなどを着用すること。また、OS 側の設定で必ずマイクがオーディオ入力機器に選択されていることを事前に確認すること。Mac の場合は、図 1 のように、「環境設定」パネルの「サウンド」で内蔵マイクを OS のサウンド入力装置として設定し、SC サーバーの起動時にリスト 1 のように、内蔵マイク入力 (Built-in Microphone) がリストの最初に列挙されているのを確認する。

リスト 1. 入出力デバイスのリスト

```
Number of Devices: 3
   0 : "Built-in Microph"
   1 : "Built-in Input"
   2 : "Built-in Output"
```

2. マイクからの入力音を得るには

マイクからの入力を SC で得るには、リスト 2 のように **SoundIn** という UGen を用いる。このプログラムでは単純にマイクからの音声をオーディオ出力装置に送っている。プログラムを実行すると、内蔵マイクに拾われた音がそのままヘッドフォンから聞こえてくる。

リスト 2. マイク入力

```
{
   SoundIn.ar([0,1]);
}.play;
```

リストでは、SoundIn の引数として [0,1] という Array が与えられているが、これは最初の 2 つのオーディオ入力チャンネル、すなわちコンピュータのステレオ・マイクの LR 両チャンネルを取得するという意味である。[0,1] の代わりに例えば「0」を与えれば、左チャンネルのみを取得することができる。

SoundIn

コンピュータのマイクやサウンドカードからの音声入力を得る。

.ar(*bus*, *mul*, *add*)

bus･･･ 入力チャンネル番号。Array を与えることで、複数のチャンネルを同時に取得可能。

3. リング・モジュレーション

まず初めに、単純なエフェクトの一例として、リング・モジュレーションをプログラムする。リング・モジュレーションは原音の周波数成分の周囲にサイドバンドと呼ばれる周波数成分を作り出して入力音の音色を変化させるもので、リスト 3 のように、SoundIn からの入力音にサイン波を掛けあわせてプログラムする事ができる。

リスト 3. リング・モジュレーション

```
{
  SoundIn.ar([0,1]) * SinOsc.ar(MouseX.kr
      (20,880));
}.play;
```

リストでは、MouseX を用いてカーソルの画面上の位置でサイン波の周波数をコントロールできるようにしている。

4. ディストーション

オーディオ信号を増幅し歪ませる事で、豊かな倍音を含んだ太い音を得る、ロック・ギターでお馴染みのディストーションは SC3 ではリスト 4 のように.distort メソッドを用いてプログラムする事ができる。

リスト 4. ディストーション

```
{
  SoundIn.ar([0,1], 100).distort * 0.1;
}.play;
```

.distort を用いて音歪ませるには、ある程度の振幅のある入力音が必要となる。リストでは、ゲインを SoundIn の *mul* 引数を 100 に設定して 100 倍に増幅し、.distort により歪ませた後、「* 0.1」で聴取しやすいよう、レベルの調整をしている。.distort の他にも.softclip を使うと.distort と違った歪み方をさせることが可能である。図 2 は 10 倍に増幅したサイン波をそれぞれ.softclip と.distort で歪ませた時の波形の違いを図示したものである。



図 2. .distort(左) と.softclip(右) により歪ませた正弦波

5. ピッチ・シフター

入力音のピッチを自在に変化させるピッチ・シフターは **PitchShift** という UGen を用いれば簡単にプログラム可能である。PitchShift はごく短いバッファに入力音を録音し、それを *pitchRatio* パラメータで指定された速度で再生することにより入力音の音程を変化させている。リスト 5 では、このパラメータを 0.5 に設定しているため、原音が半分の速度での再生される、このため原音を 1 オクターブ低くシフトされた音が作られる。

リスト 5. ピッチシフター

```
{
  PitchShift.ar(SoundIn.ar([0,1]),
      pitchRatio:0.5);
}.play;
```

例えば、原音より短三度高い音と長三度低い音を付加してピッチシフターを利用した和音を作る事も可能である。このように音階上にピッチシフトを行う場合は、以前に学習した.miditocps を利用する。

リスト 6. ピッチシフターによる和音

```
{
  a = PitchShift.ar(SoundIn.ar([0,1]),
      pitchRatio: 3.midicps / 0.midicps);
  b = PitchShift.ar(SoundIn.ar([0,1]),
      pitchRatio: -4.midicps / 0.midicps);
  a+b
}.play
```

.midicps は、MIDI ノートナンバーを周波数に変換するメソッドであり。「0.midicps」を実行すると、MIDI ノートナンバー「0」の周波数を得られる。MIDI ノートナンバー「0」より長三度高い、つまり半音階で 4 高い MIDI ノートナンバー「4」と、短三度低い、つまり半音階で 3 低い、ノートナンバー「-3」の周波数をそれぞれもとめ、それをノートナンバー「0」の周波数で除算することで、ノートナンバー「4」と「-3」と「0」間の周波数比、つまりどれだけ速く、あるいは遅く入力音を再生すれば、その音程が得られるかが数値として得る事ができる。これを *pitchRatio* の値として PitchShift.ar に与える事で、入力音より短三度高い音と長三度低い音を作り、ピッチ・シフターを用いた長三和音を実現する事ができる。

PitchShift

ピッチシフトを行う UGen。音程は *pitchRatio* で指定する。dispersion のパラメータを用いることで、音程のランダマイズなども可能。
.ar(*in*, *windowSize*, *pitchRatio*, *pitchDispersion*, *timeDispersion*, *mul*, *add*,)

in· · · 入力信号

windowSize· · · ウインドウ・サイズ

pitchRatio · · · ピッチ・レシオ。再生速度。

pitchDispersion · · · ピッチ分散度

timeDispersion · · · 時間分散度

6. ディレイ

本項では入力音を一定時間遅延させるディレイとその応用を数種紹介する。

6.1. シンプルなディレイ

遅延効果を SC で用いるには **DelayN** という UGen を使用する。この UGen の第 2、第 3 引数はそれぞれ、最大ディレイ・タイムとディレイ・タイムで、最大ディレイ・タイムの値を元に SC はディレイ用のバッファを用意し、それを利用してディレイ・タイムで指定された時間だけ入力音を遅延させて出力する。最大ディレイ・タイムを超えたディレイ・タイムを設定した場合、ディレイ・タイムは最大ディレイ・タイムにクリップされる (リスト 7)。

リスト 7. ディレイ

```
1 {
2   DelayN.ar(SoundIn.ar([0,1]), 0.5, 0.5)
3 }.play
```

DelayN

入力音を *delaytime* で指定した時間だけ遅延させる。
.ar(*in*, *maxdelaytime*, *delaytime*, *mul*, *add*)

in· · · 入力信号。

maxdelaytime· · · 最大ディレイ・タイム。ディレイ用のバッファの確保に使用する。

delaytime · · · ディレイ・タイム。最大ディレイ・タイムより大きい値を指定した場合、最大ディレイ・タイムにクリップされる。

6.2. フィードバック・ディレイ

遅延させた信号を振幅を弱め、もう一度 Delay.ar に入力することによって、入力音が減衰しながらも何度も繰り返される、やまびこのようなエフェクト、フィードバック・ディレイが実現できる（図 3)。このような



図 3. フィードバック・ディレイの成り立ち

フィードバック・ディレイを実現するには、リスト 8 のように **LocalIn** と **LocalOut** という UGen を用いる。リストでは、4 行目で DelayN による 0.5 秒の遅延処理と「* 0.5」による振幅の減衰が施された信号を LocalOut に送り、それが 2 行目の LocalIn によって取り出され、原音と加算されている。LocalOut の引数としてオーディオ信号の Array を渡す事で、複数のチャンネルを LocalOut に送る事も可能であり。複数のチャンネルを LocalOut に送った場合は、2 行目のように、そのチャンネル数 (= 2) を LocalIn の第 1 引数として指定する。

リスト 8. フィードバック・ディレイ

```
1 {
2   a = SoundIn.ar([0,1]) + LocalIn.ar(2);
3   d = DelayN.ar(a, 0.5, 0.5);
4   LocalOut.ar(d*0.5);
5   d
6 }.play
```

リストの 5 行目に「d」が置かれているのは、{} 内の最終行に書かれたものが自動的にオーディオ出力に送られるためである。つまり、DelayN により 0.5 秒の遅延を行い、「* 0.5」による減衰を施す前の音がヘッドフォンから聞こえてくることとなる。

6.3. ピンポン・ディレイ

フィードバック・ディレイにアレンジを加え、各ディレイ音が左右のスピーカーから交互に聞こえる、ピンポン・ディレイを作る事もできる。



図 4. ピンポンディレイ

左右のスピーカーに交互にディレイを送るには左右のスピーカー用にそれぞれ独立したフィードバック・ディレイを用意し、片方のスピーカーのフィードバック・ディレイに音を入力するタイミイングをもう片方に入力するタイミングとずらすことにより実現できる (図 6.3)。

リスト 9 は 0.25 秒ごとに左右のスピーカーから交互にディレイ音が聞こえるピンポン・ディレイのプログラム例である。ピンポン・ディレイは「1 つ」の音源がスピーカーを行き来するエフェクトであるため、入力音が 1 チャンネルである必要がある。このため、リストの 2 行目でマイクの LR チャンネルを加算し、モノラル化をしている。そして、3 行目では、入力音に 0.25 秒の遅延と「* 0.8」により振幅を減衰を施し、それを変数 d に代入している。また、4 行目で LocalIn を用いて LocalOut からの 2 チャンネルの信号を取得し、変数 l に代入。5、6 行目で L チャンネルを入力音 (a[0]) と、R チャンネル (a[1]) を 0.25 秒の遅延を伴った、d とミックスしている。それを、7、8 行目で、それぞれ個別の DelayN に入力することによって、0.5 秒さらに遅延させ、9 行目で LocalOut に送り、フィードバックを生じさせている。10 行目には [l,r+d] となっているため、左のスピーカーからは l 音、つまり 0.5、1.0、1.5、2.0... 秒後にディレイが聞こえ、右のスピーカーか r と d の音のミックスした r の音、つまり 0.25、0.75、1.25、1.75... 秒後にディレイが聞こえてくる。

リスト 9. ピンポン・ディレイ

```
1 {
2   i = SoundIn.ar(0) + SoundIn.ar(1);
3   d = DelayN.ar(i, 0.25, 0.25) * 0.8;
4   a = LocalIn.ar(2);
5   l = a[0] + i;
6   r = a[1] + d;
7   l = DelayN.ar(l, 0.5, 0.5) * 0.8;
8   r = DelayN.ar(r, 0.5, 0.5) * 0.8;
9   LocalOut.ar([l,r]);
10  [l,r+d]
11 }.play
```

6.4. マルチタップ・ディレイ

リスト 10 のように DelayN と Array と組み合わせて、リズミカルなディレイ (=マルチタップ・ディレイ) を作る事も可能である。

リスト 10. マルチタップ・ディレイ

```
1 {
2   t = 0.1;
3   a = [1, 2, 4, 6, 7] * t;
4   d = DelayN.ar(SoundIn.ar([0,1]), a, a);
5   Mix.ar(d);
6 }.play
```

このリストで変数 a はマルチタップ・ディレイのリズムパターンが定義された Array が格納されている。Array は [1, 2, 4, 6, 7] というリズムパターンに 0.1 が掛けられているので、その内容は [0.1, 0.2, 0.4, 0.6, 0.7] となる、この Array を DelayN の最大ディレイ・タイムとディレイ・タイムにそれぞれ適応している。このように引数に Array が与えられた場合、SC は自動的に DelayN を複製する (図 5)。



図 5. マルチタップ・ディレイ

ここでは 5 つの要素からなる Array が与えられたため、DelayN が 5 つ作られ、それぞれに異なった最大ディレイ・タイム及びディレイ・タイムが指定される。この全ての DelayN からの出力は 5 つの要素を持つ Array として 4 行目で変数に d に格納されている。そして、5 行目で **Mix** という UGen を使って、Array 内の全てをミックスしている。2 行目の変数 t を変更する事で、リズムを保持したまま、テンポだけを変える事が可能である。

Mix

array 内の音声信号を全て加算する

.ar(*array*)

array··· 音声信号の Array、全てのチャンネルが加算される

7. リバーブ

残響効果を入力音に加えるには、以下のように FreeVerb という UGen を利用するのが最も簡単である。

リスト 11. リバーブ

```
1 {
2   FreeVerb.ar(SoundIn.ar([0,1]), 1.0);
3 }.play
```

FreeVerb

原音にリバーブ（残響効果）を付加する UGen

.ar(*in*, *mix*, *room*, *damp*, *mul*, *add*)

in· · · 入力信号

mix· · · 原音と加工音のバランス、1 で加工音のみ、0. 原音のみ

room · · · ルームサイズ、1.0 が最大、0.0 が最小。

damp · · · 高周波数のダンプ。0 から 1 の範囲で指定

8. フィルタ

SC には様々なフィルタが予め用意されている。リスト 12 では、入力音にハイパス・フィルタを施し、3000Hz 以下の低周波成分を減衰させている。

リスト 12. ハイパス・フィルタ

```
1 {
2   HPF.ar(SoundIn.ar([0,1]), 3000);
3 }.play
```

SC には HPF(High Pass Filter) の他にも LPF(Low Pass Filter)、BPF(Band Pass Filter)、Moog、TwoPole など様々なフィルタが用意されている。「Filter」をヘルプで調べると、全てのフィルタのリストを見ることができる。

HPF

2 次バターワース・ハイパスフィルタ

.ar(*in*, *freq*, *mul*, *add*)

in· · · 入力信号

freq· · · カットオフ周波数

9. ワウワウ

リスト 13 では、バンド・パスフィルタ中央周波数のパラメータを LFO でコントロールすることにより、ファンク・ギターなどでお馴染みの「ワウワウ」のエフェクトを作成している。

リスト 13. ワウワウ

```
1 {
2   l = SinOsc.ar(5, 0, 1000, 1500);
3   BPF.ar(SoundIn.ar([0,1]), l, 0.5);
4 }.play。
```

BPF の三番目の引数は rq(reciprocal of Q) 値であり、この値が少なければ少ないほどフィルタを通過する帯域幅が狭くなり、強くワウワウがかかる。

BPF

2 次バターワース・バンドパスフィルタ

.ar(*in*, *freq*, *rq*, *mul*, *add*)

in· · · 入力信号

freq· · · 中央周波数

rq· · · reciprocal of Q 値

10. エフェクタを SYNTHDEF としての定義する

SC3 はこれまでに制作してきたようなエフェクタを複数組み合わせ、SynthDef として定義することも可能である。リスト 14 ではピッチ・シフター・ディレイ・リバーブを連結して、より複雑なエフェクトをプログラムし、それを「myEffect」という SynthDef として定義している。

リスト 14. エフェクトの組み合わせ

```
1 SynthDef("myEffect",{
2   i = SoundIn.ar([0,1]);
3   i = PitchShift.ar(i, pitchRatio:0.5) + i
      ;
4   i = DelayN.ar(d, 0.5, 0.5) + i;
5   i = FreeVerb.ar(d, 1.0) + i;
6   Out.ar(0, i);
7 }).load;
8
9 Synth("myEffect")
```

リストでは 3 から 5 行目にかけて、「+ i」が行末に書かれているが、これはエフェクトにより加工された音と、エフェクトへの入力音を足し、両方の音が聞こえるようにするためである。

11. まとめ

今回は SC による様々なエフェクトのプログラミングを学習した。エフェクタは、さらに SC のバス (Bus) と言われる機能を利用して、エフェクタの Synth を複数繋ぐ、音を生成する Synth とエフェクタを定義した Synth を組み合わせて、自作シンセ音に各種エフェクトを適応するという等、様々な応用が可能である。このバス機能については、また回を改めて紹介する。

12. 参考文献

[1] *SuperCollider*, http://supercollider.sourceforge.net(アクセス日 2014 年 6 月 10 日)

13. 著者プロフィール

13.1. 美山 千香士 (Chikashi Miyama)

作曲家、電子楽器創作家、映像作家、パフォーマー。国立音楽大学音楽デザイン学科より学士・修士を、スイス・バーゼル音楽アカデミーよりナッハ・ディプロムを、アメリカ・ニューヨーク州立バッファロー大学から博士号を取得。Prix Destellos 特別賞、ASCAP/SEAMUS 委嘱コンクール 2 位、ニューヨーク州立大学学府総長賞、国際コンピュータ音楽協会賞を受賞。2004 年より作品と論文が国際コンピュータ音楽会議に 13 回入選、現在までに世界 19 カ国で作品発表を行っている。 2011 年、DAAD(ドイツ学術交流会) から研究奨学金を授与され、ドイツ・カールスルーエの ZKM で客員芸術家として創作活動に従事。近著に「Pure Data-チュートリアル & リファレンス」(Works Corporation 社) がある。現在ドイツ・ケルン音楽大学講師。スイス・チューリッヒ芸術大学コンピュータ音楽・音響研究所 (ICST) 研究員。バーゼル造形大学のプロジェクト「Experimental Data Aesthetics」プログラマ。`http://chikashi.net`